# $Id: PKGBUILD 184031 2013-05-01 21:41:04Z eric $
# Maintainer:

pkgname=perl-libwww
pkgver=6.05
pkgrel=1
pkgdesc="The World-Wide Web library for Perl"
arch=('any')
url="https://metacpan.org/release/libwww-perl"
license=('PerlArtistic' 'GPL')
options=('!emptydirs')
depends=('perl' 'perl-encode-locale' 'perl-file-listing'
         'perl-html-parser' 'perl-http-cookies' 'perl-http-daemon'
         'perl-http-date' 'perl-http-negotiate' 'perl-lwp-mediatypes'
         'perl-net-http' 'perl-uri' 'perl-www-robotrules'
         'perl-http-message')
optdepends=('perl-lwp-protocol-https: for https:// url schemes')
source=("http://search.cpan.org/CPAN/authors/id/G/GA/GAAS/libwww-perl-${pkgver}.tar.gz")
sha1sums=('e10b097a4a2f04efcb3549b6543e16534b46064d')

build() {
    cd libwww-perl-${pkgver}
    perl Makefile.PL --aliases INSTALLDIRS=vendor
    make
}

check()  {
    cd libwww-perl-${pkgver}
    make test
}

package() {
    cd libwww-perl-${pkgver}
    make DESTDIR="$pkgdir" install
}
